
ВСТУП 
Біологія продуктивності с.-г. тварин - наука про біологічні особливості 

онтогенезу с.-г. тварин, що дає можливість в значній мірі підвищити 
рентабельність та вихід продукції тваринництва та поєднує в собі 
фундаментальні та прикладні науки в системі біологічних дисциплін.  

Біологія продуктивності пов’язана з біохімією та фізіологією, що дає 
можливість встановити напрямок обміну речовин з метою виділення 
першочергових продуктів необхідних для харчування тварин. Годівля, селекція, 
генетика та ембріологія встановлює процес життєдіяльності осіб, що базується 
на головних факторах взаємодії тварин з довкіллям. Зв’язок біології 
продуктивності з імунологією дозволить підвищити стійкість тварин до 
негативної дії патогенних мікроорганізмів.  

Біологія продуктивності тварин пов’язана з молекулярною генетикою, з 
ДНК аналізом, що дозволяє визначити біля 4000 генетичних хвороб та 
схильність до захворювання, на інфекційні онкологічні, та інформує про 
генеалогію та родинні зв’язки. Біологія продуктивності тісно пов’язана з 
генетикою та розведення с.-г. тварин, що є теоретичною основою для 
розв’язання практичних задач по відтворенню та селекції тварин. 

Мета дисципліни:  освоєння здобувачами основних закономірностей 
фізіологічних та біохімічних показників, метаболізму, що визначає онтогенез 
тварин та детермінує молочну, м’ясну, вовнову, яєчну продуктивності, 
медоносність, тощо. 

Завдання дисципліни:   
• вивчення фізіолого-біологічних механізмів регуляції молочної, 

м ясної, яєчної, вовнової, хутрової, шкурної, медової та ін. продуктивності с.-г. 
тварин; 

• вивчення біології розвитку та визначення етапу життєвого циклу, 
під час якого одержується найбільша кількість високоякісної продукції при 
мінімальних витратах ресурсів; 

• визначення основних принципів створення паратипових умов, що 
забезпечують їх високу продуктивність та обумовлюють виникнення у 
нащадків нових властивостей чи ознак; 

• визначення особливостей різних типів поведінки тварин (харчової, 
шлюбної тощо), а також стресостійкості для забезпечення їх високих 
продуктивних якостей; 

• створення генотипів з бажаними властивостями з використанням 
сучасних методів маркер-залежної селекції. 

Предмет дисципліни: вивчення закономірностей фізіологічних та 
біохімічних показників, метаболізму, що визначає онтогенез тварин та 
детермінує молочну, м’ясну, вовнову, яєчну продуктивності, медоносність. 

Об’єкт дисципліни: фізіологічні та біохімічні показники, метаболізм, 
онтогенез тварин та механізми детермінації молочної, м’ясної, вовнової, яєчної 
продуктивності, медоносність бджіл, тощо. 

Практична робота  № 1-2 
 
Тема: Автоматизація аналізу варіаційного ряду 
Мета: Набути навичок створення алгоритму автоматизованого аналізу 

варіаційного ряду даних первинного зоотехнічного обліку 
 

1. Відкрийте файл “Basa-data1.xls”. Він містить дані первинного 
зоотехнічного обліку для 390 голів корів, а також показники продуктивності 
предків. Провести аналіз змінної “Надій матерів”; вся вихідна інформації 
міститься на аркуші “Лист 1” в інтервалі С4:С393.  

Насамперед необхідно надати власне ім’я цього аркушу та створити 
новий аркуш, на якому буде розташовуватися таблиця варіаційного ряду.   Для 
цього, клацніть правою кнопкою миші (ПКМ) по назві “Лист1”; у меню, що 
з’явиться обрати опцію “Переименовать”. Назва “Лист1” буде виділено білим 
кольором на чорному фоні. З клавіатури набрати “База даних” та клацнути 
кнопку Enter. 

Для того, щоб створити новий аркуш необхідно:  клацнути по опції 
головного меню “Вставка” та обрати опцію “Вставить Лист”. Новий аркуш, 
що з’явиться, буде мати назву “Лист1”. Змінити його ім’я на “Гістограма”.  

2. Для того, щоб побудувати варіаційний ряд попередньо необхідно 
визначити кількість інтервалів та ширину інтервалу. При цьому розрахунок цих 
вибіркових ознак повинно залежати як від вихідних даних, так й від обсягу 
вибірки. При зміні кількості значень у вибірки чи зміні будь-якого значення, всі 
параметри варіаційного ряду повинні змінюватися автоматично, без втручання 
людини.  

Для цього на аркуші “База 
даних” у його клітках створить 
таблицю наступного виду. 

Розрахуємо необхідні для 
побудови варіаційного ряду 
допоміжні величини. Для цього у 
клітку С396 введемо:  

=СЧЁТ (С4:С393).  
Це можна зробити або з 

клавіатури (для цього перейти на 
російський алфавіт), або через 
опцію головного меню  “Вставить 

функцию” f x .  
У останньому випадку у клітку С396 введіть знак дорівнює =, потім  

клацніть по кнопці f x , відкриється “Мастер функций”. Далі у лівому віконці 
оберіть категорію “Статистические”, а після цього у правому - функцію 

“СЧЁТ”; клацніть по кнопці ОК. У вікні, що відкриється наберіть С4:С393 
(тільки з англійського алфавіту!) та клацніть ОК. У клітки С396 повинно 
з’явитися цифра 390.  

3. У клітки С397 набрати =МИН(С4:С393), а у клітки С398 – 
=МАКС(С4:С393).  Або знову ж використати “Мастер функций”; в категорії 
“Статистические” обрати функції “МИН” та “МАКС”. В обох випадках 
введіть область розрахунку С4:С393. Якщо після цього у базі даних з’явиться 
значення ще більше чи ще менше, ніж у поточній базі даних, значення кліток 
С397 (або С398) зміниться. Знаходимо розмах варіювання; у клітки С399 
наберемо =С398-С397. 

4. Далі розраховуємо кількість інтервалів варіаційного ряду. Згідно 
формули Старджеса це значення залежить від обсягу вибірки: i = 1+3,32·lg n.  

Але при цьому воно повинно бути цілим числом та непарним. Тому у 
клітки С400 наберемо =НЕЧЁТ(ЦЕЛОЕ(1+3,32*LOG10(C396))) .  

5. Розрахуємо ширину інтервалу. Для цього у клітку С401 вводимо 
=ОКРВВЕРХ(C399/C400;10). Таким чином, ми задаємо наступним режим 
обрахунку: показник розмаху варіювання (клітка С399) буде віднесено до 
кількості інтервалів (клітка С400), а отримане значення округлене до 
найближчого більшого числа, кратного 10. Оскільки 9473/9 = 1052,56, в даному 
випадку це значення буде дорівнювати 1060. 

Остання дія підготовчого етапу – це розрахунок нижньої межі першого 
інтервалу варіаційного ряду. Для цього мінімальне значення (клітка С397) 
необхідно округлити до найменшого кратного краще всього в даному випадку 
50. Тому у клітку С403 вводимо =ОКРВНИЗ(C397;50) . 

6. Для зручності таблицю варіаційного ряду та гістограму фактичного та 
теоретичного нормального розподілів розташуємо на окремому аркуші – на 
аркуші “Гістограма”, який ми вже створили раніше. Для того, щоб перейти з 
аркушу “База даних” на аркуш “Гістограма” просто клацніть лівою кнопкою 
миші (ЛКМ) по назві аркуша – “Гістограма”. На цьому чистому аркуші 
необхідно побудувати наступну таблицю: 

 
 

7. Розраховуємо нижні межі кожного інтервалу. Оскільки ці значення 
відрізняються між собою на величину інтервалу, можна використати наступну 
формулу для розрахунку значень нижніх інтервалів (Xmin, i ):  

Xmin, i = (i – 1)·h + Xmin, 1, 

де  і – номер інтервалу, а h – ширина інтервалу, Xmin, 1 – нижня межа першого 
інтервалу. Номера інтервалів наведено у клітках А4:А12 аркушу “Гістограма”, 
ширина інтервалу у клітці С401 аркушу “База даних”, а нижня межа першого 
інтервалу – у клітці С403 аркушу “База даних”.  

Тому у клітку В4 аркушу “Гістограма” вводимо: 
=((A4-1)*'База даних'!$C$401+'База даних'!$C$403)+1 

та клацаємо Enter.  
Для того, щоб заповнити клітки В5:В12, візьміться курсором миші за 

нижній правий ріг клітки В4 та розтягніть його донизу таблиці. Цей стовпчик 
таблиці заповниться автоматично розрахованими значеннями нижньої межі 
відповідних інтервалів.  

Для розрахунку верхньої межі першого інтервалу у клітку С4 аркушу 
“Гістограма” введіть:  

=B4-1+'База даних'!$C$401. 
Для того, щоб заповнити інші клітки цього стовпчику – знову 

використайте операцію розтягування формули, як у попередньому випадку. 
(В даному випадку при розрахунку нижньої межі ми додавали до 

розрахункового значення ще 1, яка тут грала роль найменшої одиниці 
вимірності. Тому нижня межа кожного наступного інтервалу на 1 більше, ніж 
верхня межа кожного попереднього інтервалу. При аналізу вмісту жиру 
найменшою одиницею вимірності буде 0,01 %.) 

8. Розрахунок середин інтервалів. Для розрахунку середини першого 
інтервалу необхідно у клітку D4 ввести вираження: =(B4-1+C4)/2. Для 
розрахунку середин наступних інтервалів знову розтягнути формулу з клітки 
D4 вниз по всьому стовпчику. 

9. Для розрахунку частот варіаційного ряду у клітку Е4 введіть формулу: 
=ЧАСТОТА('База даних'!$C$4:$C$393;C4). 

Згідно цій формулі програма буде підраховувати кількість значень у 
інтервалі С4:С393 на аркуші “База даних”, які будуть менше, ніж верхня межа 
першого інтервалу (у даному випадку – менше, ніж 1860). Якщо цю формулу 
розтягнути на у клітки Е5:Е12, то ми отримаємо накопичені частоти інтервалів 
варіаційного ряду. У самій нижній клітці накопичена частота дорівнюватиме 
390, тобто всі варіанти вибірки повинні бути менше, ніж верхня межа 
останнього інтервалу ряду.  

Абсолютні частоти варіаційного ряду можна розрахувати на підставі 
накопичених частот. Для першого інтервалу завжди абсолютна частота 
дорівнює накопиченій; таким чином у клітку F4 необхідно ввести =Е4. 
Абсолютна частота будь-якого іншого інтервалу дорівнює різниці між 
накопиченою частотою цього інтервалу та накопиченою частотою 
попереднього  інтервалу. Тому у клітку F5 вводимо =E5-E4. Для заповнення 
інших кліток цього стовпчику розтягуємо цю формулу донизу. 

10. Далі, для перевірки відповідності розподілу нашого варіаційного  
нормальному розподілу нам необхідно розрахувати теоретичні частоти (як 

абсолютні, так й накопичені). Тому першим ділом розраховуємо для кожного 
інтервалу нормоване відхилення.  

Для цього у клітку G4 вводимо у одну строку формулу: 
=(D4-СРЗНАЧ('База даних'!$C$4:$C$393))/СТАНДОТКЛОН('База 
даних'!$C$4:$C$393). 

Клацаємо Enter. Далі розтягуємо цю формулу донизу стовпчика.  
У клітку Н4 вводимо формулу: =НОРМРАСП(G4;0;1;ЛОЖЬ); й 

розтягуємо цю формулу вниз стовпчика. Програма розрахує нам ординати 
нормальної стандартизованої кривої (з середнім арифметичним рівним 0, то 
сигмою рівною 1) для значення нормованого відхилення кожного інтервалу.  

Для того, щоб від стандартизованої нормальної кривої перейти до 
фактичної, необхідно розрахувати теоретичні абсолютні частоти. Для цього 
необхідно ординати нормальної кривої помножити на обсяг вибірки та на 
ширину інтервалу й  віднести до вибіркової сигми. Тому у клітку І4 вводимо в 
один рядок наступну формулу: 

 
=H4*'База данных'!$C$396*'База данных'!$C$401/ СТАНДОТКЛОН('База 

данных'!$C$4:$C$393)  

та клацаємо Enter. Розтягуємо клітку І4 вниз стовпчика.  
11. Розраховуємо теоретичні накопичені частоти. Накопичена частота 

першого інтервалу дорівнює його абсолютній частоті. Тому у клітку J4 
вводимо вираз: =І4 . Накопичена частота будь-якого іншого інтервалу 
дорівнює сумі накопиченої частоти попереднього інтервалу та абсолютній 
частоті цього. Тому у клітку J5 вводимо: =J4+I5. Розтягнемо цю формулу 
донизу стовпчика. 

12. Перевірка відповідності емпіричного розподілу нормальному 
проводиться за допомогою критерію Колмогорова-Смирнова. Для цього 
необхідно розрахувати абсолютні різниці накопичених частот варіаційного 
ряду та теоретичних накопичених частот по всім інтервалам. Для цього у 
клітку K4 вводимо формулу: =ABS(J4-E4) та клацаємо Enter. Розтягуємо 
клітку К4 вниз стовпчика.  

Далі необхідно визначити найбільше з цих значень. Для цього у клітку 
К16 вводимо формулу: =МАКС(K4:K12). Для того, що розрахувати значення 
критерію Колмогорова-Смирнова необхідно у клітку К19 ввести формулу: 
=K16/('База данных'!C396)^0,5 . Якщо розраховане значення буде менше, ніж 
1,36 – вибірковий розподіл не відрізняється від нормального. Якщо 
розраховане значення критерію буде перевищувати 1,36 – вибірка має 
ненормальний розподіл. В цьому випадку її не можна аналізувати класичними 
методами. 

 
13. Побудова гістограми. Для того, щоб зобразити у графічній формі 

отримані результати можна побудувати гістограму фактичного розподілу та 
теоретичного (нормального). 

Для цього необхідно створити додаткову таблицю із вихідними даними: 

 
 
В ній стовпчик Інтервал місить значення нижньої (В4-В12) та верхньої 

межі (С4-С12) кожного інтервалу, стовпчик Pi містить абсолютні частоти 
нашого варіаційного ряду, тобто F4-F12, а стовпчик Pi_Teo – відповідні 
теоретичні частоти (тобто, I4-I12). 

Далі повністю виділяємо блок В15: D24, в головному меню обираємо 
«Вставка» - «Гистограмма» - «Гистограмма с группировкой» та на аркуші 
з’явиться гістограма фактичного розподілу та теоретичного. 

14. В підсумку аркуш “Гістограма” повинен мати наступний вигляд: 

Практична робота № 3 
 
Тема: Автоматизація біометричної обробки даних зоотехнічного обліку 
Мета: Набути навичок створення алгоритму автоматизованого 

розрахунку біометричних показників  
 

1. Відкрийте файл “Basa-data1.xls”. Створити новий аркуш за назвою 
“Біометрія”.  Для цього клацніть ЛКМ на кнопку Головного меню “Вставка” 
та обрати опцію “ та обрати опцію “Вставить Лист”. В книзі з’явиться новий 
аркуш з назвою “Лист 1”. Змінити його ім’я на “Біометрія”. 

 
2. На аркуші “Біометрія” створити наступну таблицю: 

 
 
3. Розрахунок блоку “Загальні характеристики вибірки”. Ці показники 

вже розраховані нами на аркуші “База даних”, тому ми просто їх переносимо. 
Для цього введемо: 

- у клітку В3: ='База даних'!C396 . 
- у клітку В4: ='База даних'!C397 . 
- у клітку В5: ='База даних'!C398 . 
- у клітку В6: ='База даних'!C399 . 
4. Розрахунок блоку “Показники центру розподілу”. В цьому блоці 

наведено найбільш поширені та важливі середні (як ступеневі, так й 
структурні).  

Для розрахунку середнього арифметичного у клітку В8 введемо формулу: 
=СРЗНАЧ('База даних'!C4:C393). Це можна зробити або з клавіатури, або (що 
краще), використавши можливості “Мастера функций”. В останньому випадку 
виконайте наступні кроки: 

а) з клавіатури наберіть у клітці В8 знак “=”; 
б) клацніть по піктограмі Головного меню fx; 
в) у вікні меню “Мастера функций”, що з’явиться, оберіть категорію 
“Статистические” (в лівому віконці) та функцію “СРЗНАЧ” (в правому 
віконці; клацніть ОК; 
г) кнопкою “Delete” очистіть віконце “Число 1”; 
д) клацніть по назві аркушу “База даних”; 
є) у віконці “Число 1” після адреси аркушу ‘База даних’! введіть з 
клавіатури адресу інтервалу вихідних даних – С4:С393; 
ж) клацніть по кнопці ОК; у клітці В8 з’явиться цифра – 6372,6461538… 
Для розрахунку медіани у клітину В9 введіть наступну формулу: 

=МЕДИАНА('База даних'!C4:C393). Або, краще, використайтесь 
можливостями “Мастера функций”, та виконайте всі кроки, як у 
попередньому випадку, але на кроці в) в категорії “Статистические” оберіть 
функцію “МЕДИАНА”. Всі наступні кроки також без змін. В клітці В9 повинна 
з’явитися цифра – 6377,5. 

Для розрахунку модального значення у клітину В10 (з клавіатури чи за 
допомогою “Мастера функций”) введіть формулу:                      =МОДА('База 
даних'!C4:C393) . 

Нарешті, для розрахунку 5% урізаного середнього арифметичного у 
клітці В11 наберіть: =УРЕЗСРЕДНЕЕ('База даних'!C4:C393;0,05) . Для цього 
виконайте всі кроки, як у попередніх випадках, але після вводу у віконце 
“Число 1” адреси аркушу та адреси вихідного інтервалу, після цифри 393 
надрукуйте знак “;” та число “0,05”. Це означає, що буде розраховане середнє 
арифметичне на вибірці, у якій 2,5% найменших та 2,5 % найбільших значень 
не буде включено у розрахунок. 

5. Розрахунок блоку ”Показники мінливості”. Для розрахунку середнього 
відхилення у клітку В13 введіть формулу:  

=СРОТКЛ('База даних'!C4:C393) . 
Для розрахунку середнього квадратичного відхилення (сигми) у клітку 

В14 введіть формулу: =СТАНДОТКЛОН('База даних'!C4:C393) . 
Для розрахунку варіанси у клітку В15 введіть формулу: 

=ДИСП('База даних'!C4:C393) . 
Для розрахунку дисперсії вибірки (суми квадратів відхилень кожної варіанти 
вибірки від її середнього арифметичного) у клітку В16 введіть: 

=КВАДРОТКЛ('База даних'!C4:C393) . 
Нарешті, для того, щоб розрахувати коефіцієнт варіації у клітці В17 

наберіть з клавіатури наступну формулу: =(B14/B8)*100 .  

6.  Розрахунок блоку “Показники форми розподілу”. Розрахуємо спочатку 
коефіцієнт асиметрії; для цього у клітку  В19 введіть наступну формулу: 
=СКОС('База даних'!C4:C393) . 

В клітку В20 введіть формулу: =ЭКСЦЕСС('База даних'!C4:C393) .  
7. Розрахунок блоку “Структурні показники”. Для розрахунку вибіркової 

оцінки першої квартилі введіть у клітку В22 формулу: =КВАРТИЛЬ('База 
даних'!C4:C393;1) . Аналогічно, для розрахунку третьої квартилі у клітку В23 
введіть формулу: 

=КВАРТИЛЬ('База даних'!C4:C393;3) . 
Останній – третій – показник цього модулю нам буде потрібен при 

розрахунку довірчого інтервалу для медіани. Для розрахунку h у клітку В24 
введіть з клавіатури наступну формулу: 

=ОКРВВЕРХ((B3-1-1,96*(B3)^0,5)/2;1)  . 
8. Розрахунок блоку “Статистичні помилки”. Для розрахунку 

статистичної помилки середнього арифметичного необхідно в клітку Е3 ввести 
наступну формулу: =B14/КОРЕНЬ(B3)  . Це можна зробити або з клавіатури, 
або використавши можливості “Мастера функций”.  В останньому випадку 
виконайте наступні кроки: 

а) в клітині Е3 наберіть знак “=”; 
б) клацніть по клітці В14; 
в) наберіть знак “/”; 
г) на Головному Меню клацніть по кнопці “Мастера функций”; 
д) оберіть категорію “Математические”; 
є) оберіть функцію “КОРЕНЬ” та клацніть ОК; 
ж) у віконці “Число” наберіть В3 та клацніть ОК; у клітці Е3 з’явиться 
значення помилки середнього арифметичного (58,05477). 
Для розрахунку помилки медіани у клітці Е4 наберіть формулу:      

=(B23-B22)/КОРЕНЬ(B3) . Для цього введіть “=(В23-В22)”, а далі повторіть 
кроки з г) по ж), як у попередньому випадку. 

Для розрахунку помилки середнього квадратичного відхилення (сигми) у 
клітці Е5 наберіть формулу (з клавіатури та використавши “Мастер 
функций”): =B14/КОРЕНЬ(2*B3) . 

Для розрахунку помилки варіанси у клітці Е6 наберіть наступну 
формулу: =B15*КОРЕНЬ(2/B3) . 

Для розрахунку помилки коефіцієнта варіації введіть у клітці Е7 
наступну формулу: =B17/КОРЕНЬ(2*B3) . 

Для розрахунку помилки коефіцієнта асиметрії у клітку Е8 введіть 
наступну формулу: =КОРЕНЬ(6/(B3+3)) . 

Для розрахунку помилки коефіцієнту ексцесу у клітку Е9 введіть 
наступну формулу: =КОРЕНЬ(24/(B3+5)) . 

9. Розрахунок блоку “Перевірка форми розподілу”. В цьому блоці ми 
розраховуємо одновибірковий t-критерій Ст’юдента для перевірки вірогідності 
відхилення оцінок коефіцієнтів асиметрії та ексцесу нулю.  

Для розрахунку критерію Ст’юдента у випадку оцінки коефіцієнтів 
асиметрії та ексцесу введіть, відповідно, у клітку Е11 наступну формулу: 
=ABS(B19/E8) , а у клітку Е12: =ABS(B20/E9) . 

У тих випадках, коли ми перевірюємо будь-яку статистичну гіпотезу, 
бажано було би мати табличне (критичне) критерію або зразу ж відмічати, якщо 
отримане значення є вірогідним. Тому введемо табличне значення критерію 
Ст’юдента для рівня значущості p = 0,05, який в даному випадку має число 
ступенів свободи df = n – 1; для цього у клітку Е13 вводимо наступну формулу: 
=СТЬЮДРАСПОБР(0,05;B3-1). Функція, зворотна розподілу Ст’юдента 
розташована також у категорії “Статистические” “Мастера функций”.  (Це 
табличне значення критерію Ст’юдента нам ще знадобиться при розрахунку 
довірчих меж середнього арифметичного.) 

В клітині Е14 розташуємо число ступенів свободи:=B3-1 . 
10. Розрахунок блоку “Довірчій інтервал”. Для розрахунку нижньої та 

верхньої межі 95 % довірчого інтервалу середнього арифметичного значення 
необхідно, відповідно, ввести у клітку Е17 формулу: B8-E13*E3 , а у Е18: 
=B8+E13*E3. 

Для розрахунку довірчого інтервалу медіани нам знадобиться величина, 
яку ми розрахували раніше у клітці В24 (h = 176), тому що, в даному випадку, 
нижньою межею довірчого інтервалу медіани буде 176-те найменше значення у 
нашій вибірці, а верхньою межею – 176-те найбільше значення у вибірці.   

Тому, для розрахунку нижньої та верхньої межі 95 % довірчого інтервалу 
медіани, відповідно, введемо у клітку Е20 формулу: 

=НАИМЕНЬШИЙ('База даних'!C4:C393;B24), 
 а у клітку Е21:  

=НАИБОЛЬШИЙ('База даних'!C4:C393;B24) . 
Для розрахунку довірчого інтервалу для середнього квадратичного 

відхилення (сигми) ми використаємо розподіл Хі-квадрат. Для того, щоб 
розрахувати нижню та верхню межу 95 % довірчого інтервалу сигми необхідно 
ввести, відповідно, у клітку Е23 формулу: 

=B14*КОРЕНЬ(E14/ХИ2ОБР(0,025;E14)), 
а у клітку Е24:  

=B14*КОРЕНЬ(E14/ХИ2ОБР(0,975;E14)) . 
11. Таким чином, в підсумку аркуш “Биометрия”  повинен мати 

наступний вигляд: 



 14 

 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 
 



 15 

Практична робота № 4 
 
Тема: Автоматизація оцінки бугаїв-плідників за нащадками 
Мета: Набути навичок створення алгоритму автоматизованої оцінки 

бугаїв-плідників за продуктивними якостями нащадками  
 
 

1. Відкрити свій файл на аркуші “База даних”. Клацнути ЛКМ по назві 
заголовку таблиці первинних даних “Кличка Б”; цей стовпчик містить клички 
бугаїв-плідників – батьків корів, які розглядаються у аналізі. 

Для того, щоб проаналізувати скільки бугаїв-плідників було використано 
та отримати повний список їх кличок, клацніть по опції Головного меню 
“Данные”,  у меню, що з’явиться оберіть опцію “Фильтр”. У верхньому 
правому куті клітки з назвою “Кличка Б” з’явиться сірий квадратик з чорним 
трикутником. Клацніть по цьому трикутнику й у віконці з’явиться перелік 
кличок бугаїв-плідників:  

 
 
Якщо необхідно обрати з загальної бази даних лише дочок одного 

певного бугая, то клацніть спочатку по закладці «(Выделить все)» (зникнуть 
«галочки» проти клички всіх бугаїв), в цьому списку оберіть його кличку та 
клацніть в відповідному квадратику ЛКМ, а далі вже ОК.  

Вихідна база даних трансформується у базу даних, яка буде містити лише 
тих корів, які мають одного обраного батька. 

Оберемо для порівняльного аналізу бугаїв-плідників за продуктивними 
якостями нащадків (за надоєм та вмістом жиру в молоці по кращій лактації) 
лише тих 16 бугаїв, які мають більше всього нащадків. 

Для цього: 
2. Створить новий аркуш, якому надати ім’я “Бугаї”. На цьому аркуші 

нарисуйте наступну таблицю. 

3. Для того, щоб підрахувати кількість дочок для кожного бугая, наберіть 

у клітці В4 наступну формулу: 
=СЧЁТЕСЛИ('База даних'!$B$4:$B$393;A4) . 

Далі, клацніть по клітці В4 ЛКМ, візьміться курсором за малий чорний 
квадратик, що знаходиться у правому нижньому куті цієї клітки та розтягніть 
формулу донизу – до клітки В19. Відпустить ЛКМ й у всіх клітках з’являться 
відповідні числа. 
4. Для того, щоб розрахувати середній надій дочок для кожного бугая-плідника, 
по-перше, розрахуємо сумарну кількість надоєного від всіх його дочок молока, 
а потім віднесемо цю кількість до чисельності дочок. Для цього у клітку С4 
введіть наступну формулу: 

=СУММЕСЛИ('База даних'!$B$4:$B$393;A4; 'База даних'!$Z$4:$Z$393)   
(В цієї формулі в інтервалі B4:B393 розташовані індивідуальні клички бугаїв, а 
в інтервалі Z4:Z393 – показники надою дочок за кращій лактацією.) 

Розтягніть цю формулу донизу – й всі клітки цього стовпчика заповняться 
відповідними цифрами. 

Для того, щоб розрахувати середній надій у клітку D4 введіть просту 
формулу: =C4/B4 та розтягніть цю формулу донизу. 

У клітку D21 введіть середню продуктивність по стаду в цілому; для 
цього наберіть наступну формулу: =СРЗНАЧ('База даних'Z4:Z393) . 

5. Далі проаналізуємо рівень продуктивності нащадків окремих бугаїв-
плідників у порівнянні із середнім показником по стаду. 

Для цього у клітку Е4 введіть наступну формулу: =D4-$D$21. Розтягніть 
цю формулу вниз до кінця стовпчика. З’являться цифри  - позитивні  чи 
негативні, в залежності від того, переважають дочки цього бугая середній 
показник по стаду чи поступаються йому, відповідно. 

6. Оцінимо відносний рівень продуктивності дочок – нащадків обраних 
бугаїв-плідників. Для цього визначимо на скільки відсотків вони переважають 

(або поступаються) середнім показникам по стаду. Для цього у клітку F4 
введіть наступну формулу:  =E4*100/$D$21. Розтягніть цю формулу донизу.  

7. Далі, для того, щоб заповнити клітини таблиці у відношенні вмісту 
жиру в молоці, знову повторіть кроки 4-6, але для стовпчика АВ4:АВ393 
вихідної бази даних, де занесені дані щодо вмісту жиру в молоці дочок за 
кращою лактацією. 

8. Таким чином, в підсумку аркуш “Бугаї”  повинен мати наступний 
вигляд: 

 
 
9. Для того, щоб візуалізувати отримані результати, побудуємо додаткову 

таблицю, куди перенесемо дані щодо відносних відхилень дочок різних бугаїв-
плідників як за надоєм, так й вмістом жиру в молоці. 

Далі, необхідно виділити ЛКМ блок А26 : С42, клацнути по опції 
Головного меню «Вставка» та обрати формат зображення «Графік» - «График 
с маркерами»: 

 
та клацнути ЛКМ. 
 
Наслідком буде поява відповідного графіка: 

 
На ньому можна відмітити нащадків бугаїв, які суттєво переважають 

середнє по стаду як за надоєм, так й за вмістом жиру (наприклад, нащадки 
Гуланта чи Яна), або, навпаки, поступаються ньому (наприклад, нащадки Луча).  

Практична робота № 5 
 
Тема: Автоматизація розрахунків коефіцієнтів успадковуваності та 

повторюваності 
Мета: Набути навичок створення алгоритму автоматизованого 

розрахунку коефіцієнтів успадкування та повторюваності 
 
 

1. Відкрити свій файл та створити новий аркуш з назвою “Кореляція”. На 
цьому аркуші створити наступну таблицю: 

 
 

2. У В6:В11 цієї таблиці вести обсяги відповідних даних. Так, у В6 
заносимо наступну формулу:  

=СЧЁТ('База даних'!H4:H393) . 
У В7-В11 заносимо аналогічні формули, змінюємо лише адреси 

інтервалів вихідних даних. (Наприклад, для Матерів Батьків це інтервал 
М4:М393 і т.і.)  

3. Далі, розрахуємо коефіцієнти парної лінійної кореляції між 
показниками надою та вмістом жиру по кожній генерації тварин дослідного 
стада. Для цього, наприклад,  у С6 для Матерів матерів (ММ) заносимо 
наступну формулу: 

=КОРРЕЛ('База даних'!H4:H393; 'База даних'!J4:J393) . 
В даному випадку вона містить наступні дані: 

'База даних'!H4:H393 – інтервал, що містить дані щодо надою ММ; 

'База даних'!J4:J393 – інтервал, що містить дані щодо вмісту жиру в 
молоці ММ. 

Аналогічно, змінюючи адресацію інтервалу вихідних даних, 
розраховуються коефіцієнти кореляції і для інших генерацій. 

4. Для розрахунку помилки коефіцієнту кореляції між надоєм та вмістом 
жиру в молоці ММ у D6 заносимо формулу: 

=КОРЕНЬ((1-C6^2)/(B6-2)) . 
Для того, щоб  розрахувати оцінки помилок коефіцієнтів кореляції й для 

інших генерацій цю формулу можна розтягнути до низу стовпчика. 
5. Далі розрахуємо вірогідність отриманих вибіркових оцінок 

коефіцієнтів кореляції. Для цього розрахуємо значення критерію Ст’юдента та 
рівень його вірогідності.  

У Е6 заносимо формулу: =ABS(C6/D6). У F6 заносимо формулу: 
=СТЬЮДРАСП(E6;B6-2;2).  

Для розрахунку цих показників для тварин інших генерацій ці обидві 
формули можна розтягнути. 

6. Розрахуємо довірчий інтервал для вибіркових оцінок коефіцієнтів 
кореляції. Для цього використаємо z-перетворення Фішера. 

У G6 заносимо: =ФИШЕР(C6). Далі розтягнемо цю формулу донизу. У 
Н6 заносимо формулу для розрахунку помилки перетворення Фішера: 
=1/КОРЕНЬ(B6-3). Цю формулу також розтягнемо донизу. 

Для розрахунку нижньої межі 95% довірчого інтервалу у І6 заносимо 
формулу:=ФИШЕРОБР(G6-1,96*H6) .  

Для розрахунку верхньої межі 95% довірчого інтервалу у J6 заносимо 
формулу: =ФИШЕРОБР(G6+1,96*H6) .  

7. Друга частина роботи присвячена розрахунку коефіцієнтів 
успадковуваності та повторюваності по надою. Використовуємо три пари 
генерацій: ММ-М, МО-М та М-Л.  

Для розрахунку коефіцієнтів успадковуваності використовується 
формула h2 = 2·rМД, де rМД – значення коефіцієнту кореляції між показниками 
матерів та їх дочок.  

Для оцінки вибіркових значень коефіцієнтів повторюваності 
використовуємо дані по лактаціям Дочок: по 1-ій, 3-ій та кращій. Коефіцієнт 
повторюваності розраховується, як коефіцієнт кореляції між надоєм: по 1-ій та 
по Кращій лактаціям, по 1-ій та по 3-ій лактаціям, і по 3-ій та по Кращій 
лактаціям дочок. 

 8. По-перше необхідно визначити кількість парних значень, для яких 
буде розраховуватися значення коефіцієнтів кореляції. Наприклад, для пари 
ММ-М у В16 заносимо: =МИН(B6;B8) . В цій формулі перша адреса – це 
клітка, в яку занесений обсяг вибірки по ММ, а в другій – обсяг вибірки  по М. 

Аналогічно заповнюємо клітки В17 та В18, а також В23:В25.  
9. Повторюємо кроки 2-6 для розрахунку відповідних значень 

коефіцієнтів кореляції у С16:J18 (для розрахунків коефіцієнтів 
успадковуваності) та С23:J25 (для розрахунків коефіцієнтів повторюваності).   

Використовуємо при цьому наступні адреси вихідних даних на аркуші 
“База даних”: 

Адресація Ознака 
С4:С393 - надій матерів; 
Е4:Е393 - вміст жиру в молоці матерів; 
Н4:Н393 - надій матерів матерів; 
J4:J393 - вміст жиру в молоці матерів матерів; 

М4:М393 - надій матерів батьків; 
О4:О393 - вміст жиру в молоці матерів батьків; 
S4:S393 - надій по 1-й лактації дочок; 
T4:T393 - вміст жиру в молоці по 1-й лактації дочок ; 

W4:W393 - надій по 3-й лактації дочок; 
X4:X393 - вміст жиру в молоці по 3-й лактації дочок ; 
Z4:Z393 - надій по кращій лактації дочок; 

AB4:AB393 - вміст жиру в молоці по кращій лактації дочок.  
10. Для розрахунку нижньої межі довірчого інтервалу для коефіцієнту 

успадковуваності використаємо логічну функції “Якщо”. Для цього у К16 
заносимо: =ЕСЛИ(F16<0,05;2*I16;0) .  

(Вона працює наступним чином: ЯКЩО значення рівня вірогідності 
коефіцієнту кореляції менше, ніж 0,05, то у К16 занести результат множення 2 
на значення нижньої межі довірчого інтервалу коефіцієнта кореляції (клітка 
І16), а ЯКЩО НІ, то занести нуль.)  

Аналогічним чином розрахуємо й верхню межу довірчого інтервалу: 
=ЕСЛИ(F16<0,05;2*J16;0) .  

Обидві формули розтягнемо донизу. 
11. Довірчі інтервали для коефіцієнтів повторюваності розрахуємо, як 

описано в п. 10, але без множення відповідних значень коефіцієнтів кореляції 
на 2. 

12. В підсумку цей аркуш повинен мати наступний вигляд: 

Практична робота № 6 
 
Тема: Автоматизація розрахунків коефіцієнтів лінійної регресії  
Мета: Набути навичок створення алгоритму автоматизованого 

розрахунку коефіцієнтів лінійної регресії 
 
 

1. Відкрити свій файл та створити новий аркуш з назвою “Регресія”. 
Створити на цьому аркуші наступну таблицю: 

 
 

2.  Розрахуємо коефіцієнти лінійної регресії між показниками надою 
матерів та дочок по кращій лактації.  

Для цього, по-перше, необхідно розрахувати по обом рядам вихідних 
даних суми та суми квадратів. А, по-друге, суму добутків між кожною парою 
значень обох рядів вихідних даних. 

Таким чином, у В4 заносимо наступну формулу:  
=СУММ('База даних'!Z4:Z393) , 

а у С4: =СУММ('База даних'!C4:C393) . 

Для розрахунку суми квадратів заносимо наступні формули: 

- у В5: =СУММКВ('База даних'!Z4:Z393) , 

- у С5: =СУММКВ('База даних'!C4:C393) . 

Нарешті, для розрахунку суми добутків у В6 заносимо формулу: 

=СУММПРОИЗВ('База даних'!Z4:Z393; 'База даних'!C4:C393) . 
3.  У В8 переносимо кількість пар відповідних значень з аркушу 

“Біометрія”: ='Біометрія'!B3. 
 
4. Розрахуємо коефіцієнти регресії за відповідними формулами: 

для яких відповідні значення суми, суму квадратів та суму добутків ми 
розрахували вище.  

Тому заносимо у В11 наступну формулу: 
=(B8*B6-B4*C4)/(B8*C5-(C4^2))  , 

а у В12: 
=(B4*C5-B6*C4)/(B8*C5-(C4^2)) . 

 
5. Для розрахунку помилок коефіцієнтів регресії необхідно розрахувати 

залишкову варіансу за формулою: 

Таким чином, у В14 занесемо формулу: 
=(B5-B12*B4-B11*B6)/(B8-2) . 
Тоді використовуємо для розрахунку помилок коефіцієнтів регресії 

наступні формули: 

й заносимо у С11 формулу: =КОРЕНЬ(B14/(C5-((C4^2)/B8))), а у С12: 
=C11*КОРЕНЬ(C5/B8). 

6.  Розрахуємо рівень вірогідності отриманих значень вибіркових 
коефіцієнтів регресії, для цього розрахуємо значення критеріїв Ст’юдента та їх 
рівень значущості. 

Для цього у D11 заносимо наступну формулу: =ABS(B11/C11), а у D12: 
=ABS(B12/C12) . 

У Е11 заносимо формулу: =СТЬЮДРАСП(D11;B8-3;2), а у Е12 – 
аналогічну формулу, лише замінюємо D11 на D12 . 

7. Для зручності, у клітки В16:Е16 переносимо відповідні оцінки для 
коефіцієнта кореляції (з аркушу “Кореляція” значення з інтервалу С18:F18).  

8. Розрахуємо довірчі інтервали для коефіцієнту успадковуваності. (Для 
розрахунку коефіцієнту успадковуваності використовуємо формулу:     h2 = 2· 

bМД, де bМД – коефіцієнт кута нахилу лінії регресії ознаки дочок залежно від 
ознаки їх матерів. 

Для розрахунку нижньої довірчої межі 95% довірчого інтервалу заносимо 
у В19 наступну формулу:  

=ЕСЛИ(E11<0,05; 2*(B11-1,96*C11);0) . 
Для розрахунку верхньої довірчої межі 95% довірчого інтервалу заносимо 

у С19 формулу:  
=ЕСЛИ(E11<0,05;2*(B11+1,96*C11);0) . 
9. В підсумку даний аркуш повинен мати наступний вигляд: 

Практична робота № 7 
 
Тема: Автоматизація алгоритму автоматизованого дисперсійного аналізу 
Мета: Набути навичок створення алгоритму автоматизованого 

дисперсійного аналізу 
 
1. Відкрийте свій файл та створить новий аркуш із назвою 

“Дисперсійний аналіз”.  
Побудуйте на цьому аркуші наступну таблицю: 

 
 

2. По-перше перенесемо з аркушу “Кореляції” обсяги кожної з груп.  Для 
цього, наприклад, у В3 заносимо: =Кореляції!B6. Аналогічно заповнюємо 
C3:F3, лише змінюючи адресацію кліток. 

3. Розрахуємо суми. Для цього, наприклад, у В4 заносимо формулу: 
=СУММ('База даних'!H4:H393) . Аналогічно заповнюємо C4:F4, змінюючи 
адресацію кліток. 

4. У В5 заносимо формулу: =B4^2/B3 . Розтягнемо цю формулу вправо до 
F5. 

5. Розрахуємо суми квадратів. Для цього, наприклад, у В6 заносимо 
наступну формулу: =СУММКВ('База даних'!H4:H393). Аналогічно заповнюємо 
C6:F6, змінюючи адресацію кліток. 

6. Нарешті, розрахуємо суми сум по всім строкам. Для цього у G3 
заносимо формулу: =СУММ(B3:F3) , яку розтягуємо донизу таблиці. 

7. Далі, для розрахунку факторіальної, випадкової й загальної сум 
квадратів дисперсійного комплексу необхідно розрахувати допоміжну 
величину Н. Для цього у В9 занесемо формулу:  =(G4^2)/G3 . 

8. У В8 заносимо кількість градацій дисперсійного комплексу: 
=СЧЁТ(B3:F3) . 

9. Наступна частина роботи присвячена безпосередньому проведенню 
дисперсійного аналізу. Для цього заповнюємо таблицю дисперсійного аналізу.  

У стовпчик “Сума квадр.” заносимо наступні формули: 
у В13 : =G5-B9 ; у В16: =G6-B9 ; у В14: =B16-B13 . 

У стовпчик “Число ст. св.” заносимо наступні формули: 
у С13: =B8-1; у С16: =G3-1 ; у у С14: =C16-C13 . 

У стовпчик “Средн. квадрат» заносимо наступні формули: 
у D13: =B13/C13 ; у D14: =B14/C14 .  

10. Розрахуємо значення критерію Фішера. Для цього у Е13 заносимо 
наступну формулу для розрахунку дисперсійного відношення: =D13/D14 . 

Далі, визначимо його рівень значущості; для цього у F13 заносимо 
формулу: =FРАСП(E13;C13;C14) . 

11. Нарешті, після того як дисперсійний аналіз проведено, визначаємо 
силу впливу фактору. Для цього у G13 заносимо формулу: =B13/B16, а у G14 
формулу: =B14/B16 . 

12. Оскільки обсяги наших груп були не однакові, необхідно розрахувати 
оцінку сили впливу фактора з урахуванням нерівності цих груп. Для цього 
заповнюємо третю таблицю аркушу. 

У В20 розрахуємо середній зважений обсяг груп; для цього заносимо 
наступну формулу:  

=(G3-((СУММКВ(B3:F3)/G3)))/(B8-1). 
Далі розраховуємо корегований середній квадрат; у В21 заносимо 

формулу: =(D13-D14)/B20 . 
Нарешті, розраховуємо виправлену силу впливу фактора; у В22 заносимо 

формулу: =B21/(D14+B21). Його статистичну помилку B23 розраховуємо за 
формулою: =(1-B22)*C13/C14 . 

13. Таким чином, у підсумку цей аркуш повинен мати наступний вигляд: 

Практична робота № 8 
 
Тема: Автоматизація порівняльного аналізу в ряду “предки-нащадки” 
Мета: Набути навичок створення алгоритму автоматизованого 

порівняльного аналізу за рівнем продуктивності в ряду “предки-
нащадки” 

 

1. Відкрити свій файл та створити новий аркуш, який назвати “Парні 
порівняння”.  
Побудуйте на цьому аркуші наступну таблицю: 

 
 
На цьому аркуші будемо проводити порівняльний аналіз за рівнем 

продуктивності (за надоєм) між різними генераціями тварин нашої дослідної 
групи. При цьому парні порівняння будемо проводити: 

- у відношенні центральної тенденції (вибіркових середніх 
арифметичних) з використанням парного критерію Ст’юдента; 

- у відношенні рівня мінливості (вибіркових варіанс) з використанням 
критерію Фішера. 

2. На початку розрахуємо вибіркові показники (обсяги вибірок, середні 
арифметичні, середні квадратичні відхилення та інші).  

3. У В3 заносимо наступну формулу:  
=СЧЁТ('База даних'!H4:H393) . 

У В4-В8 заносимо аналогічні формули, змінюємо лише адреси інтервалів 
вихідних даних. Наприклад, для Матерів Батьків це: М4:М393, й т.п.  

 
 
Замість набору цієї формули з клавіатури можна виконати наступні 

кроки: 
а) з клавіатури наберіть у В3 знак “=”; 

б) клацніть по піктограмі fx; 

в) у вікні меню “Мастера функций”, що з’явиться, оберіть категорію 
“Статистические” (в лівому віконці) та функцію “СЧЕТ” (в правому 
віконці; клацніть ОК; 
г) у віконці, що відкриється, стерти всі записи; 
д) клацнути по імені аркуша “База даних”; 
є) після тексту 'База даних'!, що появиться у віконці, ввести з клавіатури 
Н4:Н393 та клацнути ОК. 
4. У С3 заносимо наступну формулу: 

=СРЗНАЧ('База даних'!H4:H393) . 
У С4-С8 заносимо аналогічні формули, також змінюємо лише адреси 

інтервалів вихідних даних.  
5. У Е3 заносимо наступну формулу: 

=СТАНДОТКЛОН('База даних'!H4:H393) . 
У Е4-Е8 заносимо аналогічні формули, також змінюємо лише адреси 

інтервалів вихідних даних.  
6. Розрахуємо помилки вибіркових середніх арифметичних. Для цього у 

D3 ввести наступну формулу: =E3/КОРЕНЬ(B3). Замість вводу з клавіатури 
можна використати можливості “Мастера функций”.   

Для того, щоб розрахувати оцінки помилок для всіх рядів даних 
розтягніть формулу із D3 донизу у D4-D8. 

7. Розраховуємо вибіркові коефіцієнти варіацій; для цього введіть у F3 
наступну формулу: =E3*100/C3, та розтягніть її донизу стовпчика.  
8. Далі проведемо безпосередньо попарні порівняння й заповнимо другу 
частину цієї таблиці. 

По-перше розрахуємо абсолютні відмінності між окремими парами 
тварин. Для цього у В13  розрахуємо різницю між середнім надоєм Матерів 
матерів та Кращою лактацією дочок; це можна зробити за наступною 
формулою: =C3-C8.  

Далі, у В14 розрахуємо різницю між середнім надоєм Матерів Батьків та 
Кращою лактацією дочок: =C4-C8.  

Заповнюємо подібним чином всі клітини цього стовпчика.  
Замість того, щоб вводити формули безпосередньо можна, наприклад, 

зробити наступні кроки: 
а) введіть з клавіатури у В13 знак “=”; 
б) клацніть лівою кнопкою миші (ЛКМ) по клітці С3; 
в) введіть з клавіатури знак “-“; 
г) клацніть ЛКМ по клітці С8; 
д) клацніть по кнопці “Enter”. 

9. Процедура попарного порівняння між двома рядами повністю 
автоматизована. При цьому програма розраховує оцінку рівня вірогідності 
парного критерію Ст’юдента. 
Для порівняння між надоєм Матерів матерів та Кращою лактацією дочок у 
клітку С13 введіть наступну формулу: 

=ТТЕСТ('База даних'!H4:H393; 'База даних'!Z4:Z393;2;1) . 

Тут у скобках приведена наступна інформація: 
- адреса ряду “Надій матерів матерів”; 
- адреса ряду “Надій дочок за кращою лактацією”; 
- далі, цифра “2” означає, що обрано двосторонній критерій Ст’юдента; 
- цифра “1” наприкінці означає, що обрано модифікація тесту Ст’юдента 

для парного порівняння для пов’язаних даних. 
Замість того, щоб вводити цю формулу з клавіатури використайте можливості 
“Мастера функций”. Для цього виконайте наступні кроки: 
а) у С13 введіть “=”; 

б) клацніть по піктограмі Головного меню fx; 
в) у вікні меню “Мастера функций”, що з’явиться, оберіть категорію 
“Статистические” (в лівому віконці) та функцію “ТТЕСТ” (в правому 
віконці; клацніть ОК; 
г) у віконці “Массив1” клацніть ЛКМ по назві аркушу “База даних”, а 
далі з клавіатури наберіть інтервал  Н4:Н393; 
д) у віконці “Массив2” подібним чином введіть адресу другого інтервалу 
значень - 'База даних'!Z4:Z393; 
є) у віконці “Хвосты” клацніть 2; 
ж) у віконці “Тип” клацніть 1; далі клацніть по кнопці ОК. 
10. Подібним чином, змінюючи адреси рядів, що порівнюються, 

розрахуйте всі клітки цього стовпчика. 
11. Процедура парного порівняння за характером мінливості двох рядів 

даних також повністю автоматизовано. При цьому програма розраховує рівень 
значущості критерію Фішера (одностороннього).  

Для цього у Е13 введіть наступну формулу: 
=ФТЕСТ('База даних'!H4:H393; 'База даних'!Z4:Z393) . 

Крім того, також можна використати можливості “Мастера функций”. 
Для цього виконати наступні кроки: 
а) у Е13 введіть “=”; 

б) клацніть по піктограмі Головного меню fx; 
в) у вікні меню “Мастера функций”, що з’явиться, оберіть категорію 
“Статистические” (в лівому віконці) та функцію “ФТЕСТ” (в правому 
віконці; клацніть ОК; 
г) у віконці “Массив1” клацніть ЛКМ по назві аркушу “База даних”, а 
далі з клавіатури наберіть інтервал Н4:Н393; 
д) у віконці “Массив2” подібним чином введіть адресу другого інтервалу 
значень - 'База даних'!Z4:Z393; далі клацніть по кнопці ОК. 
12. Подібним чином, змінюючи адреси рядів, що порівнюються, 

розрахуйте всі клітки цього стовпчика. 
13. Наприкінці цей аркуш повинен мати наступний вигляд: 

